

PSPICE 仿真石英晶体振荡电路

作者：陈拓 chentuo@ms.xab.ac.cn

2011年1月1日

1. 概述

多谐振荡器常用来生成脉冲信号和时钟信号。

多谐振荡器是一种自激振荡电路。因为没有稳定的工作状态，多谐振荡器也称为无稳态电路。多谐振荡器的工作过程可以这样简述，如果一开始多谐振荡器处于 0 状态，那么它在 0 状态停留一段时间后将自动转入 1 状态，在 1 状态停留一段时间后又将自动转入 0 状态，如此周而复始，输出矩形波，所以多谐振荡器也称矩形波发生器。由于在矩形波中基波之外的谐波分量很多，多谐振荡器也因此而得名。

在频率稳定度要求高的场合通常采用石英晶体振荡电路构成多谐振荡器。

频率稳定度一般用频率的相对变化量 $\Delta f/f_0$ 来表示， f_0 为振荡频率， Δf 为频率偏移。频率稳定度有时附加时间条件，如一小时或一日内的频率相对变化量。

多谐振荡器通过 LC 电路选频可以构成正弦波振荡电路。在石英晶体振荡电路中，用石英晶体取代 LC 振荡电路中的 L、C 元件组成的正弦波振荡电路，其频率稳定度可高达 10^{-9} 至 10^{-11} 。

影响 LC 振荡电路振荡频率的因素主要是 LC 谐振回路的 Q 值，Q 值越高，选频特性越好，频率越稳定。一般 LC 振荡电路的 Q 值只有数百，而石英晶体的 Q 值很高，可达到数千至数万，因而石英晶体具有优良的选频性能。

2. 反馈振荡器的工作条件

振荡器是一个反馈电路，一个反馈振荡器电路必须满足三个条件：起振条件（保证接通电源后能逐步建立起振荡），平衡条件（保证进入维持等幅持续振荡的平衡状态）和稳定条件（保证平衡状态不因外界不稳定因素影响而受到破坏）。

2.1 平衡条件

反馈振荡器的组成方框图如下。




图 1 反馈振荡器的组成方框图

随着振荡幅度的增长，放大器的动态范围就会延伸到非线性区，放大器的增益将随之下降，振幅越大，增益下降越多，最后当反馈电压正好等于原输入电压时，振荡幅度不再增大而进入平衡状态。

放大器开环电压增益和反馈系数的表示式分别为：

$$\dot{A} = \frac{\dot{U}_o}{\dot{U}_i}, \quad \dot{F} = \frac{\dot{U}_f}{\dot{U}_o} \quad \text{----- (1)}$$

进入平衡状态后 $\dot{U}_i = \dot{U}_f$ ，此时根据式(1-1)可得反馈振荡器的平衡条件为：

$$\text{环路增益 } T(j\omega_{osc}) = \dot{A}\dot{F} = AF e^{j(\varphi_A + \varphi_F)} = 1 \quad \text{----- (2)}$$

式（2）又可分别写成：

$$AF = 1 \quad \text{----- (3)}$$

$$\varphi_A + \varphi_F = 2n\pi (n = 0, 1, 2, \dots) \quad \text{----- (4)}$$

式（1-3）和（1-4）分别称为反馈振荡器的振幅平衡条件和相位平衡条件。

2.2 起振条件

式（2）是维持振荡的平衡条件，是针对振荡器进入稳态而言的。为了使振荡器接通直流电源后能够自动起振，则要求反馈电压在相位上与放大器输入电压相同，在幅度上则要求 $U_f > U_i$ ，即：

$$AF > 1 \quad \text{----- (5)}$$

$$\varphi_A + \varphi_F = 2n\pi (n = 0, 1, 2, \dots) \quad \text{--- --- (6)}$$

式（5）和（6）分别称为反馈振荡器的起振振幅条件和起振相位条件。

2.2 稳定条件

振荡器在工作过程中，不可避免地要受到各种外界因素变化的影响，如电源电压波动、温度变化、噪声干扰等。这些不稳定因素将引起放大器和回路的参数发生变化，结果使 $T(\omega_o)$ 或 $\varphi_T(\omega_o)$ 变化，破坏原来的平衡条件。

2.2.1 振幅稳定条件

要使振幅稳定，振荡器在其平衡点必须具有阻止振幅变化的能力，这就要求在平衡点附近 $T(\omega_o)$ 随 $T(\omega_o)$ 的变化率为负值，即：

$$\left. \frac{\partial T(\omega_0)}{\partial U_i} \right|_{U_i=U_{iA}} < 0$$

这就是振幅稳定条件。其中 $U_i = U_{iA}$ 是平衡点。

2.2.2 相位的稳定条件

为了保证相位稳定，要求振荡器的相频特性 $\varphi_T(\omega_0)$ 在振荡频率点应具有阻止相位变化的能力，这就要求 $\varphi_T(\omega_0)$ 曲线在 ω_0 附近应为负斜率，即：

$$\left. \frac{\partial \varphi_T(\omega)}{\partial \omega} \right|_{\omega=\omega_0} < 0$$

这就是相位的稳定条件。

3. LC 三点式正弦波振荡器

图 2 是 LC 三点式振荡器原理电路。




图 2 三点式振荡器原理电路

其中 X_{be} 、 X_{ce} 和 X_{bc} 均为电抗原件。可以看出， X_{be} 、 X_{ce} 和 X_{bc} 构成了决定振荡频率的并联谐振回路，同时也构成了正反馈所需的反馈网络。

在不考虑晶体管电抗效应的情况下，振荡频率近似等于回路的谐振频率。那么，在回路处于谐振状态时，回路呈纯阻性，有：

$$X_{be} + X_{ce} + X_{bc} = 0$$

由上式可见， X_{bc} 必须与 $X_{be}(X_{ce})$ 为性质相反的电抗元件。

根据图中规定的正方向，放大器的输出电压与其输入电压反相，即 $\varphi_A(\omega_0) = \pi$ ，而反馈电压又是输出电压在 X_{bc} ， X_{be} 支路中分配在 X_{be} 上的电压，即：

$$\dot{U}_f = \frac{jX_2}{j(X_2 + X_3)} \dot{U}_o = -\frac{X_2}{X_1} \dot{U}_o$$

为了满足相位平衡条件， $\varphi_F(\omega_0) = \pi$ 。

由上面的分析可知，在三点式电路中，L C回路中与发射极相连接的两个电抗元件必须为同性质，另外一个电抗元件必须为异性质。这就是三点式电路组成的相位判据，或称为三点式电路的组成法则。

与发射极相连接的两个电抗元件同为电容时的三点式电路，称为电容三点式电路，也称为考毕兹（Colpitts）电路，图 3。与发射极相连接的两个电抗元件同为电感时的三点式电路，称为电感三点式电路，也称为哈特莱（Hartley）电路，图 4。




图 3 电容三点式振荡器原理电路




图 4 电感三点式振荡器原理电路

4. 电源为电压源时 LC 振荡电路的相关公式

4.1 LC 串联谐振电路

图 5 是 LC 串联谐振电路。




图 5 LC 串联谐振电路及其电抗特性曲线

串联谐振电路的谐振频率：
$$f_s = \frac{1}{2\pi\sqrt{LC}}$$

品质因数定义：电感器或电容器在谐振时产生的电抗功率与电阻器消耗的平均功率之比，称为谐振时的品质因数。品质因数 Q 值愈大电路对谐振时的响应越好。

串联谐振电路之品质因数：
$$Q = \frac{1}{R}\sqrt{\frac{L}{C}}$$

4.2 LC 并联谐振电路

图 6 是 LC 并联谐振电路。




图 6 LC 并联谐振电路及其电抗特性曲线

并联谐振电路的谐振频率： $f_s = \frac{1}{2\pi\sqrt{LC}}$

并联谐振电路之品质因数： $Q = R\sqrt{\frac{L}{C}}$

5. 石英晶体振荡器的等效线路及参数

石英晶体（Quartz Crystal）的外观图片实例见图 7。



图 7 一些石英晶体样品的照片

5.1 石英晶体的代表符号、等效电路和电抗特性

将石英晶体振荡器转换成振荡频率附近的 Butterworth-Van Dyke (BVD)等效电路。在这个等效电路中，主要有四个主要参数：静态电容 C_0 (Static Capacitance，也称并联电容 Shunt Capacitance)，动态电容 C (Motional Capacitance)，动态电感 L (Motional Inductance) 及动态电阻 R (Motional Resistance，也称等效串联电阻 Equivalent Series Resistance)。下图为石英晶体的符号、等效电路和电抗特性。




图 8 石英晶体的符号、等效电路和电抗特性

由等效电路可知，石英晶体有两个谐振频率，即：

- (1) L-C-R 支路串联谐振频率 f_s (series resonance frequency)

$$f_s = \frac{1}{2\pi\sqrt{LC}}$$

- (2) 并联谐振频率 f_p (parallel resonance frequency)

当 $f > f_s$ 时，L-C-R 支路呈感性，与 C_0 产生并联谐振。

$$f_p = f_s \sqrt{1 + (C / C_0)}$$

注：总电容是电容 C 个 C_0 的并联： $\frac{CC_0}{C + C_0}$

因此：


$$f_p = \frac{1}{2\pi\sqrt{L \cdot \frac{CC_0}{C + C_0}}} = \frac{1}{2\pi} \sqrt{\frac{C + C_0}{LCC_0}} = \frac{1}{2\pi\sqrt{LC}} \sqrt{1 + \frac{C}{C_0}} = f_s \sqrt{1 + (C / C_0)}$$

表 1 是典型的石英晶体参数，通过观察这些数据对石英晶体的参数会有初步的认识。数

据引自仙童（Fairchild）公司的应用注释 Application Note 340（AN340）。

表 1 典型石英晶体参数

TABLE 1. Typical Crystal Parameters



Parameter	32 kHz fundamental	200 kHz fundamental	2 MHz fundamental	30 MHz overtone
R ₁	200 kΩ	2 kΩ	100 Ω	20 Ω
L ₁	7000H	27H	529 mH	11 mH
C ₁	0.003 pF	0.024 pF	0.012 pF	0.0026 pF
C ₀	1.7 pF	9 pF	4 pF	6 pF
Q	100k	18k	54k	100k

可以看出 $C_0 \gg C_1$ ，所以 $f_p \approx f_s$ 。

5.2 晶体振荡器传输响应的测量




图 9 性晶体振荡器传输响应的测量

照片中是 8MHz 晶体振荡器的传输响应，右图是 HP3585 频谱/网络分析仪的抓图。被测试的晶体清楚地显示出一个串联谐振点，在 8,000,250Hz（COUNTER 8 000 250.1 Hz）处，这一点有最小阻抗，对应于最大传输幅度。接下来，大约在频率高出 12kHz 处，有一个并联谐振点，这一点有最大阻抗，对应于最小传输幅度。然后是一些更高频率的响应，但幅度较低，这些谐振是由其他晶体振动模式产生的寄生谐振，通常与机械间断有关，我们通常不关心这些谐振，因为振荡器将被设计为锁定在主谐振上。

5.3 电路工作在 f_s 还是在 f_s 与 f_p 之间

现在我们知道利用石英晶体的谐振频率有 f_s 与 f_p ，我们怎样确定电路工作在哪个谐振频率呢？

5.3.1 在石英晶体上串联一个电容

在实际应用中，通常串入一个用于校正振荡频率的小电容 C_s ，如图 10 所示。




图 10 串入一个用于校正振荡频率的小电容 C_s

新的谐振频率为：

$$f_N = \frac{1}{2\pi\sqrt{LC}} \sqrt{1 + \frac{C}{C_0 + C_s}} = f_s \sqrt{1 + \frac{C}{C_0 + C_s}}$$

由此看出： $C_s \rightarrow 0$ 时 $f_N = f_p$ ； $C_s \rightarrow \infty$ 时 $f_N = f_s$ 。

调节 C_s 可以使 f_N 在 f_s 和 f_p 之间变化。

1) 当石英晶体发生串联谐振时，频率在 f_s ，它呈纯阻性，相移是 0。若把石英晶体作为放大电路的反馈网络，并起选频作用，只要放大电路的相移也是 0，则满足相位条件，形成串联型石英晶体正弦波振荡电路。

2) 当石英晶体频率在 f_s 与 f_p 之间，石英晶体呈感性，可将它与两个 C 构成电容三点式正弦波高频振荡电路，形成并联型石英晶体正弦波振荡电路。

图 11 是石英晶体各谐振点的位置。




图 11 石英晶体各谐振点的位置

5.3.2 在石英晶体上并联一个电容

如图 12，在石英晶体两端并联一个电容，通常为 30pF 左右。




图 12 并联一个用于校正振荡频率的小电容 C_L

$$f_N = f_s \left[1 + \frac{C}{2(C_0 + C_L)} \right]$$

5.3.3 负载电容和频率牵引

在石英晶体上串联或并联的电容都可以看作负载电容（Load Capacitance），负载电容对频率的影响如图 13 所示。




图 13 a) 单独的晶体元件 b) 晶体元件和负载电容 C_L 串联 c) 晶体元件与负载电容 C_L 并联

负载电容是从晶体的两个引脚向电路系统看去电路所呈现的全部有效电容，它与晶体元件一起决定晶体在电路上的工作频率，用负载电容改变晶体频率称为晶体频率牵引。

无论是负载电容与晶体元件串联还是并联，负载电容对负载谐振频率的影响都是相同的，用下式能计算相对负载谐振频率偏移：

$$\Delta_L = \frac{F_N - F_S}{F_S} = \frac{C}{2(C_0 + C_L)}$$

5.4 石英晶体的标称频率

晶体外壳所标注的频率通常称为标称频率，它既不是串联谐振频率也不是并联谐振频率，而是在外接负载电容时测定的频率，数值介于串联谐振与并联谐振频率之间。通常通过负载电容校准，使电路工作在标称频率。

6. 石英晶体振荡器

6.1 石英晶体振荡器的电路结构

将石英晶体引入普通多谐振荡器就能构成具有很高频率稳定性的石英晶体多谐振荡器。多谐振荡器就是矩形波发生器，矩形波可以分解成无穷多个正弦波分量，只有频率为石英晶体谐振频率的正弦波分量可以通过石英晶体，形成正反馈，而其它正弦波分量无法通过石英晶体，通过石英晶体的正弦波分量被反相器转换成频率相同的矩形波。

用逻辑门电路和石英晶体能构成振荡频率很稳定的脉冲波形，通常用在数字逻辑电路中，作为时钟脉冲信号。

石英晶体振荡器有两种电路结构：串联型晶体多谐振荡器和并联型石英晶体振荡器。

6.1.1 串联型晶体多谐振荡器

图 14 是串联型晶体多谐振荡器电路图。




图 14 串联型晶体多谐振荡器电路图

将谐振频率选在串联谐振频率 f_s 处，串联谐振时电抗为 0，电路呈纯阻性，晶体设置在正反馈回路，构成串联型石英晶体振荡器。

在串联型晶体多谐振荡器电路中，R 是偏置电阻，使反相器 G1 的静态时工作点落在电压传输特性曲线的线性区，以利起振，晶体串在正反馈回路。电路中 V_{O1} 和 V_{O2} 近似为正弦，经 G3 整形后变成方波。

6.1.2 并联型晶体多谐振荡器

在并联型晶体多谐振荡器中最常见的是三点式晶体振荡器，石英晶体以电感的形式出现在电路中。根据交流接地方法的不同，三点式晶体振荡器又可以分为皮尔斯（Pierce）振荡器、科尔皮兹（Colpitts）振荡器和克拉普（Clapp）振荡器，它们的结构如图 15 所示。




图 15 皮尔斯振荡器、科尔皮兹振荡器和克拉普振荡器

它们都是电容三点式电路。相比较皮尔斯电路较好，因此在实际应用中多采用它。
图 16 是常用的并联型晶体多谐振荡器电路图，是皮尔斯电路。




图 16 并联型晶体多谐振荡器电路图

该电路的工作频率并不在谐振点上，而选在 f_s 和 f_p 之间，使晶体呈现电感性，以便形成电容三点式振荡。

实际使用中以并联石英晶体振荡器与 COMS 单反相器组成的电路（Pierce Crystal Oscillator 皮尔斯石英晶体振荡器）最为简单、常用。许多 MCU 都用这种形式，一般反相器和偏置电阻是集成在 MCU 中的，见图 17。




图 17 反相器 74HC04 和石英晶体振荡器组成的谐振电路

反相器 U1A 与负反馈偏置电阻 R 组成单级反相放大器，相移为 180° 。这种由 CMOS 反相器和石英晶体组成的振荡电路具有相当稳定的振荡频率，一般可达 $10^{-8} \sim 10^{-10}$ 数量级。

6.2 基本原理

其实这个电路就是典型的电容三点式振荡电路，X1 是晶体，作为电路中的反馈元件，相当于三点式里面的电感（石英晶体工作在 f_s 与 f_p 之间时相当于一个大电感），C1 和 C2 就是电容，74HC04 和大电阻 R 实现 CMOS 反向放大电路，反相放大器工作在线性放大状态，当电路振荡频率接近石英晶体的串联谐振频率时，满足振荡条件。大家可以参考高频电子线路书里的三点式电容振荡电路进行分析。

在这里 74HC04 不是作为数字电路的反相器使用,而是作为反向模拟放大器使用。所以, 74HC04 必需要并联一个电阻, 不然它将处于饱和区或截止区, 而不是放大区。R 相当于三极管的偏置作用, 让 74HC04 处于放大区域, 这时 74HC04 就是一个反相器放大器。

让输入的直流电位保持在恰好让输出处于 H 和 L 电平的中间状态, 就可以当反向放大器来用了: 用 1 只 1-10 M 的电阻即可实现。用 CMOS 反相器, 输入阻抗高, 相比晶体管电路更容易稳定振荡。

接下来用通俗的方法讲解一下这个三点式振荡电路的工作原理。大家知道一个正弦振荡电路要振荡的条件是, 系统放大倍数大于 1, 这个容易实现; 并且相位满足 360° 。

下面讲讲这个相位问题:

74HC04 因为是反相器, 也就是说实现了 180° 移相, 那么就需要 C1, C2 和 Y1 再实现 180° 移相就可以, 恰好, 当 C1, C2, Y1 形成谐振时, 能够实现 180 移相, 这个大家可以解方程等, 把 Y1 当作一个电感来做。也可以用电容电感的特性, 比如电容电压落后电流 90° , 电感电压超前电流 90° 来分析, 都是可以的。

6.3 C1、C2 对稳定性的影响

C1、C2 这两个电容是晶振的负载电容, 一般在几十皮法。它会影响到晶振的谐振频率和输出幅度。

当 C1 增大时, C2 端的振幅增强, 当 C2 降低时, 振幅也增强。


有些时候 C1, C2 不焊也能起振, 这并不是说没有 C1, C2, 而是因为芯片引脚的分布电容起到了 C1, C2 的作用, 因为本来 C1, C2 就不需要很大。接下来分析这两个电容对振荡稳定性的影响。

因为 74HC04 的电压反馈是靠 C1 的, 假设 C1 过大, 反馈电压过低, 不稳定; 假设 C1 过小, 反馈电压过高, 储存能量过少, 容易受外界干扰, 也会增加辐射影响外界。C2 的作用对 C1 恰好相反。因为我们布板的时候, 假设双面板, 比较厚的, 那么分布电容的影响不是很大, 假设在高密度多层板时, 就需要考虑分布电容, 尤其是 VCO 之类的振荡电路, 更应该考虑分布电容。

一般晶体在基频下最高频率只能做到 30MHz 附近, 最高的也只有 40MHz, 更高的一般采用晶体的 3 次倍频, 这个时候晶体的接法比较特殊, 需要采用一定的选频网络, 图 18, 这个选频网络需要工作在 3 次倍频上, 这样才能保证稳定工作。但一般还是建议采用基频晶体, 所以客户在采购高频晶体时, 需要问清楚厂家是基频还是倍频, 超过 40MHz 的大部分都是倍频了。

不同频率范围晶体的应用电路

基频 1MHz—25MHz



三次泛音: 27MHz—66MHz




图 18 不同频率范围晶体的应用电路

7. PSPICE CMOS 反相器放大器仿真

本文使用 **Capture CIS Lite Edition 9.2** 绘制电路图，用 **PSpice Lite Version 9.2** 仿真。

MOSFET 即 **Metal Oxide Semiconductor Field Effect Transistor** 金属氧化物半导体场效应晶体管。

CMOS (**Complementary Metal Oxide Semiconductor**)，互补金属氧化物半导体，电压控制的一种放大器件。是组成 CMOS 数字集成电路的基本单元。

CMOS 由 PMOS 管和 NMOS 管共同构成，由于 PMOS 与 NMOS 在特性上互补，因此称为 CMOS。CMOS 的特点是低功耗。由于 CMOS 中一对 MOS 组成的门电路在瞬间要么 PMOS 导通、要么 NMOS 导通、要么都截至，比线性的三极管(BJT)效率要高得多，因此功耗很低。

CMOS 的基本电路单元是反相器 (Inverter)。CMOS 反相器一般多用于数字电路的高低电平转换之用，而由于它具有高输入阻抗以及高反馈阻抗的特性，将它用作模拟信号放大器使用时，在某一频带范围内可得到相当高的电压增益。

7.1 CMOS 的电压传输特性介绍

CMOS 的基本组态是由一对 PMOS 和 NMOS 晶体管互补组成的非门，如图 19。CMOS 的其传输特性曲线如图 20，线性区的中点为静态工作点。图 21 是共源交流放大器。




图 19 CMOS 非门的基本结构

图 20 COM 非门的电压传输特性曲线

图 21 共源交流放大器

从 CMOS 的电压传输特性曲线可知，在高低电平转换过程中存在一个线性放大区，且与小信号放大器和运算放大器的传输特性相似，因此，只要将其工作点设置在线性工作区的中点，即 $V_{DD}/2$ 处，就可以将 CMOS 非门构成线性放大器。

一般非门构成的放大器如图 22 所示。




图 22 单级非门构成的放大器

图 19 是图 22 电路反相器的内部结构，其中 R_F 是反馈偏置电阻，用来设置静态工作点， $R_F=10M$ 时，放大器的增益约为 20~30 倍。

7.2 CMOS 反相放大器仿真

用 PSPICE 仿真图 16 左图时，不能直接用非门模型，例如 74HC04 的模型，因为非门是数字电路模型。要得到正确的仿真结果要用图 16 右图，即用 CMOS 反相放大器代替非门 74HC04。下面我们用 PSPICE 仿真 CMOS 反相放大器。

7.2.1 电路图

- 创建新项目

1. 打开 OrCAD Capture CIS Lite Edition。
2. 创建一个新项目：File > New > Project。
3. 输入项目的名字，例如 MyCMOSInverter。生成的项目文件的扩展名为.opj。
4. 选择 Analog or Mixed-AD 模拟或混合-AD。
5. 在 Location 框中输入项目路径。点击 OK。
6. 在 Create PSpice Project 对话框打开时，选择“Create Blank Project”。

- 绘制电路图

如果我们不能确定所需的元件在哪个库中，可以选中所有的库，在 Part:文本框中逐个输入元件的名字，例如 Mbrean, Capture 会自动补足名字后面的字母，并且在 Part 列表中用“元件/库”的格式指出元件所在的库，在右下显示角元件的电路图。从图中我们知道，元件 MbreakN 在 BREAKOUT 库中。




图 23 从原理图元件库中找我们需要的元件

MOSFET 模型的符号位于 breakout.olb 库中，仿真模型位于 breakout.lib 库中。NMOS 和 PMOS 的名字分别是 MbreakN 和 MbreakP。放置好 MOSFET 符号后我们需要定义 MOSFET 的参数，图 24 是 NMOSSE 的参数定义。




图 24 NMOSFET 参数定义

在这里不设置参数 L 和 W 也可以仿真，但是 MOSFET 最基本的参数，我们设置 L、W 的目的是顺便熟悉一下 MOSFET 最基本的参数。在属性表中修改参数 L 和 W，我们只要输入值即可。这两个参数默认不显示，如果想显示，例如参数 L=1.5u，可以在属性窗口中选中 L 参数，再点击 Display 按钮，修改显示属性为 Name and Value，即可显示，图 25。




图 25 改变参数的显示属性

L、W 分别是 MOS 场效应晶体管沟道的长、宽。




图 26 参数 L、W 的示意图

IDS 与 W 成正比，与 L 成反比，见下式：

$$I_{DS} = \frac{k'_n}{2} \frac{W}{L} (V_{GS} - V_{Th})^2$$

SPICE 仿真 MOSFET 参数很多，在本文中不再做过的说明，请参考有关文献。绘制完成的电路见图 27。




图 27 CMOS 反相器仿真电路图

注意：RL 的阻值是 1MEG，不要写成 1M，两者相差甚远。

- 生成网表文件

在仿真之前先要生成网表文件。执行菜单命令 PSpice > Create Netlist, 生成的网表文件, 见图 28。

```

1: * source MYCMOSINVERTER
2: M_M1      N00322 N06892 0 0 MbreakN
3: + L=1.5u
4: + W=2u
5: V_V1      N06892 0 0Vdc
6: M_M2      N00322 N06892 N06972 N06972 MbreakP
7: + L=1.5u
8: + W=2u
9: V_V2      N06972 0 5Vdc
10: R_RL      0 N00322 1MEG
11:

```

图 28 CMOS 反相器测试电路的网表文件

7.2.2 直流静态工作点仿真

1. 打开原理图, 在 PSpice 菜单上选择 New Simulation Profile。
2. 在文本框 Name 中输入一个描述性的名字, 例如 Bias。
3. 从 Inherit From 列表中选择 none 并点击 Create。
4. 当 Simulation Setting 仿真设置窗口打开时, 对于 Analysis Type 分析类型, 选择 Bias Point 偏置点并点击 OK。




图 29 直流静态工作点仿真设置

5. 运行仿真: PSpice> Run。
6. 一个状态窗口将打开, 让你知道是否仿真成功, 如果有错, 可查看仿真输出文件, 或 Session Log 窗口。
7. 为了看到直流偏置点的仿真结果, 你可以打开仿真输出文件或返回原理图并点击 V

图标（偏置电压显示）和 I 图标（偏置电流显示）显示电压和电流，见图 30。

为了检查电流方向，你必须查看网表：电流的正方向是从节点 1 流到节点 2（见上面有关电流方向的注释）。




图 30 直流静态工作点仿真结果

7.2.3 直流传输特性曲线

为了找到合适的静态工作点，对电路进行 0 和 5V 之间的电压源扫描。

1. 从 Pspice 菜单创建一个新的 New Simulation Profile 仿真配置文件；我们将称它为 Transfer characteristics, Inherit From 还是 none。
2. 我们做直流扫描 DC sweep；输入将被扫描的电压源的名字：V1，分别指定开始值、结束值和步距：0、5 和 0.01V，让 V1 从 0~5V 做线性变化，步长为 0.01V（见图 31）。




图 31 DC Sweep 的设置

- 在反相器的输入和输出端各放置一个探针，见图 32




图 32 放置测试探针

- 运行仿真 Pspice > Run。PSpice 将产生一个包含电路中所有电压和电流值的输出文件。




图 33 直流传输特性曲线

从图中可以看出，静态工作点设置在 2.5V 最合适，通常是 $V_{CC}/2$ 。

其实在这里 V1 用什么源都没有关系，后面我们将 V1 换成正弦信号源做瞬态分析时，再执行 Transfer characteristics 仿真配置文件结果是一样的，

7.2.4 瞬态分析（时域分析）

我们将 V1 换为正弦信号源做瞬态分析，偏移 VOFF 设置静态工作点为 2.5V，幅度为 0.5V，频率为 10kHz，如图 34 所示。




图 34 瞬态分析电路图

1. 如上图所示从 SOURCE 库插中选择 VSIN 信号源按图 34 进行设置。
2. 设置瞬态分析: 从菜单选择 PSpice > New Simulation Profile 命令。命名为 Transient。
3. 当仿真设置窗口打开时, 选择 Time Domain (Transient)时域瞬态分析。输入运行时间, 我们设它为 1ms。对于 Maximum Step 最大步长的大小, 你可以让它空着或输入 10us, 如果空着波形不光滑, 越小波形越光滑。
4. 运行 Pspice, 一个探针窗口将打开显示仿真结果, 见图 35。




图 35 瞬态分析仿真结果

输出波形失真是因为放大器的线性不好, 这一点从图 31 中可以看出, 可根据实际情况通过修改 MOSFET 的参数改变工作点附近的线性。

我们现在有了 3 个仿真配置文件, 见图 36, 红色的那一个是当前活动的。如果我们现在要进行 Transfer characteristics 仿真, 右击选择它, 再选择 Make Active 命令, 就可以改变当前活动的仿真配置文件为 Transfer characteristics。




图 36 红色的仿真配置文件是当前活动的

8. PSPICE 石英晶体振荡器仿真

晶体振荡器是应用最广泛的电路之一，但是由于晶体振荡器高 Q 值的原因，对于它的仿真是非常困难的。高 Q 值会导致在振荡器达到稳定状态之前有很长的启动延迟。在某些情况下，启动延迟时间太长了以至于难以进行仿真分析，本文对石英晶体振荡器的仿真进行了一些探讨。下面我们对图 17 所示的反相器和石英晶体振荡器做 PSPICE 仿真。

8.1 电路图和网表文件

参照图 17 的右图绘出如图 37 的仿真电路图。

晶体的参数 L 、 R 、 C 和 C_0 取自前面“表 1 典型晶体参数”中 2M 晶体的参数。实际参数你需要查生产商的产品手册。OSCILENT 公司的网站 <http://www.oscilent.com/> 上有许多产品手册，你可以参考。

I_{start} 是起振用的脉冲电流源，因为在 SPICE 模型中各元件都处于平衡状态，仿真时必须加一个扰动打破平衡电路才能起振。而实际电路中多少总有扰动存在，所以容易起振。




图 37 反相器和石英晶体振荡器组成的谐振电路的仿真电路图

反相器的 MOSFET 使用 MbreakN 和 MbreakP 模型，没有设置 L 和 W 参数。 R_f 是设置静态工作点的偏流电阻。

R_s 用来控制反馈强度。

C_1 和 C_2 是负载电容，其作用我们在前面介绍过。

R_L 是负载电阻。

电路绘制完成后生成网表文件。见图 38。

```
xtal_osc-schematic1.net
1: source XTAL_OSC
2: R_R      NO00031 N16241 100
3: C_C2     0 NO1882 20pF
4: M_M2     NO1882 NO0166 VDD VDD MbreakP
5: C_C      NO0166 NO00031 0.012pF
6: C_C1     0 NO0166 20pF
7: V_V1     VDD 0 5V
8: C_C0     NO0166 NO0117 4pF
9: L_L      N16241 NO0117 529mH
10: R_Rs    NO1882 NO0117 1K
11: R_Rf    NO0166 NO1882 2.2MEG
12: R_RL    NO1882 0 100MEG
13: I_Istart NO0117 N16241 DC 0Aac AC 0Aac
14: +PULSE 0 1m 0 10n 10n 50n 1
15: M_M1     NO1882 NO0166 0 0 MbreakN
16:
```

图 38 石英晶体振荡器谐振电路的网表文件

8.2 瞬态分析

添加仿真配置文件 tran，为了测得较稳定的波形，我们将波形片段取后一些，例如 10u~11u，如图 39 进行设置。




图 39 瞬态分析的仿真配置文件设置

进行仿真，结果见图 40。




图 40 石英晶体振荡器谐振电路的仿真结果

在图中利用标尺工具测量周期为 500.000nS，对应的频率为 2MHz，幅度的峰峰值为 4V 左右。由于绘图的精确度所限测量有误差，减小步长可提高精度，但延长仿真时间。

计算串联谐振频率：

$$f_s = \frac{1}{2\pi\sqrt{LC}} = \frac{1}{2\pi\sqrt{0.529 \times 0.012 \times 10^{-12}}} = 1997568.46 \text{ Hz}$$

计算并联谐振频率：

$$f_p = f_s \sqrt{1 + (C/C_0)} = 1997568.46 \times \sqrt{1 + [(0.012 \times 10^{-12}) / (4 \times 10^{-12})]} = 2000562.57 \text{ Hz}$$

在本电路中石英晶体振荡器工作在感性区，工作频率在 f_s 和 f_p 之间。

8.3 使用 XTAL 库中的石英晶体模型仿真

默认安装的 Orcad Family Release 9.2 Lite Edition 库中不带石英晶体模型 XTAL，需要从网上找到石英晶体模型的元件符号库 xtal.olb 和石英晶体模型的仿真库 xtal.lib，复制到文件夹. \Capture\Library\PSpice 下。

在 Capture CIS Lite Edition 中打开 Place Part 窗口，点击 Add Library 按钮将元件符号库 xtal.olb 添加到项目中。见图 40。

从图 41 中我们看到有 8 种石英晶体模型，元件名称代表的意义：QZ – QUARTZ, S – 串联谐振, P – 并联谐振, 频率 100K、1MEG、10MEG、32768Hz 和 3.579545MHz – QZPCBRST。




图 41 添加石英晶体元件符号库 XATL.olb

画好的电路见图 42。在图中我们没有办法象图 37 那样在晶体 X1 中添加一个起振用的脉冲电流源 Istart，但我们可以为电容 C1 设置初始电压值 IC=0 来改变电路的稳态，强迫电路从非偏压点位置开始仿真。IC 是 Initial Condition 的缩写，用来改变瞬态分析的初始值。




图 42 石英晶体振荡器仿真电路图

对该电路做直流静态工作点仿真，见图 43，可以对比静态工作点的变化。电容上的初始电压只是瞬时激发了电路的振荡，对电路的特性没有影响。




图 43 设置 IC 前后电路静态工作点的改变

图 44 是瞬态分析的设置。正如你所看到的仿真时间很长这是因为电路的启动要很长时间，如果时间设置短了，电路还未起振，你可能会以为电路不工作。




图 44 石英晶体振荡器仿真电路的瞬态分析的设置

仿真之前要手工把石英晶体的仿真库 xtal.lib 添加到项目中，见图 45。




图 45 添加石英晶体的仿真库 xtal.lib

图 46 是仿真结果，你会发现电路大约从 3.5mS 处开始才起振。




图 46 石英晶体振荡器仿真结果

其实该电路不设置 C1 的 IC 值也能起振。在图 44 中选中 Skip the initial transient bias point calculation 选项，该选项的作用是忽略瞬态偏置电的计算，直接使用各元件的起始条件进行瞬态分析。因为 Pspice 在仿真时不会把电感、电容的初始值默认当作 0，而是会根据电路计算出来，这一点从图 43 可知。

图 47 就是去掉 C1 的 IC 值后（图 43 的左图）的仿真结果。




图 47 不设置 C1 的 IC 值时石英晶体振荡器仿真结果

电路大约在 0.7mS 处就起振了。

下面我们取一个小片段做测量。图 48 是瞬态分析设置。图 49 是仿真结果。



图 48 改变瞬态分析设置以测量波形的周期和幅度




图 49 测量波形的周期和幅度

在图中利用标尺工具测量周期为 1.0000uS，对应的频率为 1MHz，幅度的峰峰值为 5V 左右，与图 47 对比，峰峰值的误差较大。

8.4 使用 XTAL 库中的石英晶体模型仿真不同频率的石英晶体

我们知道在 XTAL 库中只有 8 种石英晶体模型，如果我们需要仿真的石英晶体频率不在模型库中怎么仿真呢？我们先来看看仿真库 xtal.lib 的内容。

```
* Library of quartz crystals

* Copyright OrCAD, Inc. 1998 All Rights Reserved.

* $Revision: 1.7 $
* $Author: RPEREZ $
* $Date: 20 Apr 1998 13:57:48 $

* -----
* The parameters in this model library were derived from:
*
* Quartz-crystal timing accuracy
* Electronic Design 2, January 19, 1976; pp. 74-79
*
* and:
*
* Electronics Designers' Handbook, Second Edition
* edited by L.J. Giacoletto
* McGraw-Hill Book Co., 1977, pp. 16-16 to 16-31

* Example use: X3 3 4 QZ32768

* Notes:
* 1) Temperature variation of resonant frequency for properly cut
* crystals is quadratic. The coefficient depends on the cut used
* and is implemented in these subcircuits by giving the equivalent
* inductor a temperature coefficient. The AT cut is an exception,
* and has a cubic temperature dependence which is not included in
* these models.
* 2) The values of lqz, cs, and cp do not vary much from one crystal
* to the next. The value of rqz, however, can vary as much as a
* factor of 2 up or down. So the crystal's Q can vary over a
```

```

*   range of 4 to 1.
* 3) Parallel resonant crystals can be tuned slightly by attaching an
*   external capacitor in parallel with the crystal. The crystal's
*   frequency is built assuming a value of the external capacitor,
*   called the "calibration capacitance". Its value is noted in
*   the comment at the head of each parallel resonant crystal model.
*$
* -----
* 32768 hertz watch crystal, XY cut, series resonant, Q=81780
*
.subckt QZS32768 1 2
*
  lqz    1    11    lmod 4448.72259
  .model lmod ind(tc2 = 8.68e-8)
  cs 11   12    5.30279780e-015
  rqz    12    2     11.2k
  cp 1    2     1.84pf
.ends
*$
* -----
* 100khz frequency standard, DT cut, parallel resonant, Q=20000
*   calibration capacitance = 5pf
.subckt QZP100K 1 2
*
  lqz    1    11    lmod 588.873289
  .model lmod ind(tc2=3.5e-8)
  cs 11   12    4.30423941e-015
  rqz    12    2     18.5k
  cp 1    2     1.72169576e-012
.ends
*$
* -----
* 100khz frequency standard, DT cut, series resonant, Q=20000
.subckt QZS100K 1 2
*
  lqz    1    11    lmod 588.873289
  .model lmod ind(tc2=3.5e-8)
  cs 11   12    4.30148495e-015
  rqz    12    2     18.5k
  cp 1    2     1.72059398e-012
.ends
*$
* -----
* 1Mhz frequency standard, AT cut, parallel resonant, Q=25000,
*   calibration capacitance = 13pf
.subckt QZP1MEG 1 2
*
  lqz    1    11    2.54647909
  cs 11   12    9.95357648e-015
  rqz    12    2     640
  cp 1    2     2.48839412e-012
.ends
*$
* -----
* 1Mhz frequency standard, AT cut, series resonant, Q=25000
.subckt QZS1MEG 1 2
*
  lqz    1    11    2.54647909

```

```

cs 11 12 9.94718394e-015
rqz 12 2 640
cp 1 2 2.48679599e-012
.ends
*$
* -----
* 3.579545Mhz color burst, AT cut, parallel resonant, Q=25000,
* calibration capacitance = 18pf
.subckt QZPCBRST 1 2
*
lqz 1 11 .0555779237
cs 11 12 3.56169600e-014
rqz 12 2 50
cp 1 2 8.90424001e-012
.ends
*$
* -----
* 10Mhz frequency standard, AT cut, parallel resonant, Q=25000
* calibration capacitance = 50pf
.subckt QZP10MEG 1 2
*
lqz 1 11 2.54647909e-003
cs 11 12 9.96041181e-014
rqz 12 2 6.4
cp 1 2 2.49010295e-011
.ends
*$
* -----
* 10Mhz frequency standard, AT cut, series resonant, Q=25000
.subckt QZS10MEG 1 2
*
lqz 1 11 2.54647909e-003
cs 11 12 9.94718394e-014
rqz 12 2 6.4
cp 1 2 2.48679599e-011
.ends
*$
* -----
* end of library file

```

在每个模型中，**lqz**、**cs**、**rqz** 和 **cp** 分别对应图 8 石英晶体等效电路中的 L、R、C 和 C0。仿真其他频率石英晶体比较简单的方法就是改变这些参数，当然你也可以创建自己的模型库进行仿真。下面我们通过改变 **QZP1MEG** 的参数仿真 2M 的石英晶体。数据来自表 1，修改部分如下：

```

* -----
* 1Mhz frequency standard, AT cut, parallel resonant, Q=25000,
* calibration capacitance = 13pf
.subckt QZP1MEG 1 2
*
lqz 1 11 0.529
cs 11 12 12e-015
rqz 12 2 100
cp 1 2 4e-012
.ends

```

图 50 是仿真结果及波形测量。在图中利用标尺工具测量周期为 500.000nS，对应的频率为 2MHz。




图 50 2M 晶振仿真结果

8.5 用参数仿真分析各元件值变化对振荡器的影响

为了分析 C1 值的变化对电路的影响，我们先将 C1 的值用参数表示，见图 51。




图 51 用参数表示 C1 的值

添加一个仿真配置文件，命名为 Parametric，先进行常规设置，与 8.3 节相同见图 52。




图 52 常规设置

再进行参数扫描设置，见图 53。




图 53 参数扫描设置

图 54 是参数扫描仿真结果，可以看到，随负载电容 C1 的增大，波形的幅度减小，相位有改变。




图 54 参数扫描仿真结果

为节省篇幅，其他元件参数值变化对电路的影响在这里就不再做了，读者可自行练习。

8.6 用非门整形输出

石英晶体振荡器产生的波形是正弦波，而数字电路中的时钟需要矩形波，我们可以用非门整形输出矩形波。画电路图时需要添加 74hc.olb 符号库才可以找到 74HC 电路，见图 55。




图 55 添加 74HC 数字电路库

电路如图 56 所示。图中，74HC04 是数字电路，其他是模拟电路。




图 56 带非门整形输出的石英晶体振荡器电路

瞬态仿真设置如图 57 所示。




图 57 瞬态仿真的常规设置

仿真之前还需要添加 74hc.lib 和 dig_io.lib 库，见图 58。




图 58 添加仿真库

图 59 是模-数电路混合仿真的结果。




图 59 反相器整形输出混合电路仿真结果

图中 VOUT 是 74HC04 的输出，绿色的部分是高低电平，黄色部分是上升和下降沿。从图中可以看出，74HC04 的输出波形不是很好，我们将反相器 74HC04 换成施密特反相器 74HC14 整出的波形就好多了，见图 60。




图 60 74HC14 施密特反相器整形输出混合电路仿真结果

参考文献（未注明来源者均来自互联网）:

- [1] 高频振荡电路-石英晶体振荡高频电路原理与分析,
<http://www.zdh1909.com/html/xuekezhishi/modianshudian/85721063829.html>。
- [2] 创易讲座系列四：石英晶体、晶振介绍
- [3] LC 正弦振荡器虚拟实验 - 东南大学内部资料
- [4] 宋自恆 1 林慶仁 2, 以 CMOS 反相器作為類比信號放大器的原理, 新電子雜誌 2001·8 月號第 185 期 p177~184.。
 - 1. 品佳股份有限公司 研發工程師
 - 2. 中央研究院地球科學研究所 技士
- [5] 电子爱好者, www.imefan.com
- [6] 行鸿彦, 基于非门电路的高增益交流放大器理论研究与应用, 电子测量与仪器学报, 2002, 16(1)
- [7] HCMOS Crystal Oscillators, Fairchild Semiconductor, Application Note 340, May 1983
- [8] Let's Be Crystal Clear,
http://www.eefocus.com/%E8%AE%BE%E8%AE%A1%E6%88%91%E6%9C%80%E8%B5%9E/blog/08-10/159474_91333.html
- [9] PSpice Archive MOSFETs, Department of Electrical Engineering University of Texas at Tyler, Copyright 2007, David M. Beams
- [10] PSpice Archive Oscillators, Department of Electrical Engineering University of Texas at Tyler, Copyright 2007, David M. Beams
- [11] PSpice Reference Guide, Copyright © 1985-2000 Cadence Design Systems, Inc. All rights reserved.
- [12] 祈放, 振荡器电路 SPICE 仿真的研究与实现, <<长春师范学院学报(自然科学版)>>2004 年 第 23 卷 第 02 期